## 80C51 family architecture

### 80C51 ARCHITECTURE

### MEMORY ORGANIZATION

All 80C51 devices have separate address spaces for program and data memory, as shown in Figures 1 and 2. The logical separation of program and data memory allows the data memory to be accessed by 8-bit addresses, which can be quickly stored and manipulated by an 8-bit CPU. Nevertheless, 16-bit data memory addresses can also be generated through the DPTR register.

Program memory (ROM, EPROM) can only be read, not written to. There can be up to 64k bytes of program memory. In the 80C51, the lowest 4k bytes of program are on-chip. In the ROMless versions, all program memory is external. The read strobe for external program memory is the <u>PSEN</u> (program store enable).

Data Memory (RAM) occupies a separate address space from Program Memory. In the 80C51, the lowest 128 bytes of data memory are on-chip. Up to 64k bytes of external RAM can be addressed in the external Data Memory space. In the ROMless version, the lowest 128 bytes are on-chip. The CPU generates read and write signals,  $\overline{RD}$  and  $\overline{WR}$ , as needed during external Data Memory accesses.

External Program Memory and external Data Memory may be combined if desired by applying the RD and PSEN signals to the inputs of an AND gate and using the output of the gate as the read strobe to the external Program/Data memory.

### Program Memory

Figure 3 shows a map of the lower part of the Program Memory. After reset, the CPU begins execution from location 0000H. As shown in Figure 3, each interrupt is assigned a fixed location in Program Memory. The interrupt causes the CPU to jump to that location, where it commences execution of the service routine. External Interrupt 0, for example, is assigned to location 0003H. If External Interrupt 0 is going to be used, its service routine must begin at location 0003H. If the interrupt is not going to be used, its service location is available as general purpose Program Memory. The interrupt service locations are spaced at 8-byte intervals: 0003H for External Interrupt 0, 000BH for Timer 0, 0013H for External Interrupt 1, 001BH for Timer 1, etc. If an interrupt service routine is short enough (as is often the case in control applications), it can reside entirely within that 8-byte interval. Longer service routines can use a jump instruction to skip over subsequent interrupt locations, if other interrupts are in use.

The lowest 4k bytes of Program Memory can either be in the on-chip ROM or in an external ROM. This selection is made by strapping the  $\overline{\text{EA}}$  (External Access) pin to either V<sub>CC</sub>, or V<sub>SS</sub>. In the 80C51, if the  $\overline{\text{EA}}$  pin is strapped to V<sub>CC</sub>, then the program fetches to addresses 0000H through 0FFFH are directed to the internal ROM. Program fetches to addresses 1000H through FFFFH are directed to external ROM.

If the  $\overline{EA}$  pin is strapped to V<sub>SS</sub>, then all program fetches are directed to external ROM. The ROMless parts (8031, 80C31, etc.) must have this pin externally strapped to V<sub>SS</sub> to enable them to execute from external Program Memory.

The read strobe to external ROM, <u>PSEN</u>, is used for all external program fetches. <u>PSEN</u> is not activated for internal program fetches.

The hardware configuration for external program execution is shown in Figure 4. Note that 16 I/O lines (Ports 0 and 2) are dedicated to bus functions during external Program Memory fetches. Port 0 (P0 in Figure 4) serves as a multiplexed address/data bus. It emits the low byte of the Program Counter (PCL) as an address, and then goes into a float state awaiting the arrival of the code byte from the Program Memory. During the time that the low byte of the Program Counter is valid on Port 0, the signal ALE (Address Latch Enable) clocks this byte into an address latch. Meanwhile, Port 2 (P2 in Figure 4) emits the high byte of the Program Counter (PCH). Then PSEN strobes the EPROM and the code byte is read into the microcontroller.

Program Memory addresses are always 16 bits wide, even though the actual amount of Program Memory used may be less than 64k bytes. External program execution sacrifices two of the 8-bit ports, P0 and P2, to the function of addressing the Program Memory.



Figure 1. 80C51 Block Diagram







Figure 3. 80C51 Program Memory

### **Data Memory**

The right half of Figure 2 shows the internal and external Data Memory spaces available to the 80C51 user. Figure 5 shows a hardware configuration for accessing up to 2k bytes of external RAM. The CPU in this case is executing from internal ROM. Port 0 serves as a multiplexed address/data bus to the RAM, and 3 lines of Port 2 are being used to page the RAM. The CPU generates  $\overline{RD}$ and  $\overline{WR}$  signals as needed during external RAM accesses. There can be up to 64k bytes of external Data Memory. External Data



Figure 4. Executing from External Program Memory

Memory addresses can be either 1 or 2 bytes wide. One-byte addresses are often used in conjunction with one or more other I/O lines to page the RAM, as shown in Figure 5.

Two-byte addresses can also be used, in which case the high address byte is emitted at Port 2.

Internal Data Memory is mapped in Figure 6. The memory space is shown divided into three blocks, which are generally referred to as the Lower 128, the Upper 128, and SFR space.

## 80C51 family architecture

Internal Data Memory addresses are always one byte wide, which implies an address space of only 256 bytes. However, the addressing modes for internal RAM can in fact accommodate 384 bytes, using a simple trick. Direct addresses higher than 7FH access one memory space, and indirect addresses higher than 7FH access a different memory space. Thus Figure 6 shows the Upper 128 and SFR space occupying the same block of addresses, 80H through FFH, although they are physically separate entities.

The Lower 128 bytes of RAM are present in all 80C51 devices as mapped in Figure 7. The lowest 32 bytes are grouped into 4 banks of 8 registers. Program instructions call out these registers as R0 through R7. Two bits in the Program Status Word (PSW) select which register bank is in use. This allows more efficient use of code space, since register instructions are shorter than instructions that use direct addressing.



Figure 5. Accessing External Data Memory If the Program Memory Is Internal, the Other Bits of P2 Are Available as I/O



Figure 7. Lower 128 Bytes of Internal RAM

The next 16 bytes above the register banks form a block of bit-addressable memory space. The 80C51 instruction set includes a wide selection of single-bit instructions, and the 128 bits in this area can be directly addressed by these instructions. The bit addresses in this area are 00H through 7FH.

All of the bytes in the Lower 128 can be accessed by either direct or indirect addressing. The Upper 128 (Figure 8) can only be accessed by indirect addressing.

Figure 9 gives a brief look at the Special Function Register (SFR) space. SFRs include the Port latches, timers, peripheral controls, etc. These registers can only be accessed by direct addressing. Sixteen addresses in SFR space are both byte- and bit-addressable. The bit-addressable SFRs are those whose address ends in 0H or 8H.



Figure 6. Internal Data Memory



Figure 8. Upper 128 Bytes of Internal RAM



Figure 9. SFR Space



Figure 10. PSW (Program Status Word) Register in 80C51 Devices

### 80C51 FAMILY INSTRUCTION SET

The 80C51 instruction set is optimized for 8-bit control applications. It provides a variety of fast addressing modes for accessing the internal RAM to facilitate byte operations on small data structures. The instruction set provides extensive support for one-bit variables as a separate data type, allowing direct bit manipulation in control and logic systems that require Boolean processing.

#### **Program Status Word**

The Program Status Word (PSW) contains several status bits that reflect the current state of the CPU. The PSW, shown in Figure 10, resides in the SFR space. It contains the Carry bit, the Auxiliary Carry (for BCD operations), the two register bank select bits, the Overflow flag, a Parity bit, and two user-definable status flags.

The Carry bit, other than serving the function of a Carry bit in arithmetic operations, also serves as the "Accumulator" for a number of Boolean operations.

The bits RS0 and RS1 are used to select one of the four register banks shown in Figure 7. A number of instructions refer to these

RAM locations as R0 through R7. The selection of which of the four is being referred to is made on the basis of the RS0 and RS1 at execution time.

The Parity bit reflects the number of 1s in the Accumulator: P = 1 if the Accumulator contains an odd number of 1s, and P = 0 if the Accumulator contains an even number of 1s. Thus the number of 1s in the Accumulator plus P is always even. Two bits in the PSW are uncommitted and may be used as general purpose status flags.

#### **Addressing Modes**

The addressing modes in the 80C51 instruction set are as follows:

#### **Direct Addressing**

In direct addressing the operand is specified by an 8-bit address field in the instruction. Only internal Data RAM and SFRs can be directly addressed.

## 80C51 family architecture

#### Indirect Addressing

In indirect addressing the instruction specifies a register which contains the address of the operand. Both internal and external RAM can be indirectly addressed.

The address register for 8-bit addresses can be R0 or R1 of the selected bank, or the Stack Pointer. The address register for 16-bit addresses can only be the 16-bit "data pointer" register, DPTR.

#### **Register Instructions**

The register banks, containing registers R0 through R7, can be accessed by certain instructions which carry a 3-bit register specification within the opcode of the instruction. Instructions that access the registers this way are code efficient, since this mode eliminates an address byte. When the instruction is executed, one of the eight registers in the selected bank is accessed. One of four banks is selected at execution time by the two bank select bits in the PSW.

#### **Register-Specific Instructions**

Some instructions are specific to a certain register. For example, some instructions always operate on the Accumulator, or Data Pointer, etc., so no address byte is needed to point to it. The opcode itself does that. Instructions that refer to the Accumulator as A assemble as accumulator specific opcodes.

#### **Immediate Constants**

The value of a constant can follow the opcode in Program Memory. For example,

MOV A, #100

loads the Accumulator with the decimal number 100. The same number could be specified in hex digits as 64H.

#### Indexed Addressing

Only program Memory can be accessed with indexed addressing, and it can only be read. This addressing mode is intended for reading look-up tables in Program Memory A 16-bit base register (either DPTR or the Program Counter) points to the base of the table, and the Accumulator is set up with the table entry number.

The address of the table entry in Program Memory is formed by adding the Accumulator data to the base pointer.

Another type of indexed addressing is used in the "case jump" instruction. In this case the destination address of a jump instruction is computed as the sum of the base pointer and the Accumulator data.

#### **Arithmetic Instructions**

The menu of arithmetic instructions is listed in Table 1. The table indicates the addressing modes that can be used with each instruction to access the <br/>byte> operand. For example, the ADD A,<br/>byte> instruction can be written as:

ADD	a, 7FH	(direct addressing)
-----	--------	---------------------

ADD	A, @R0 (indirect addressing)
ADD	a, R7 (register addressing)

ADD A, #127 (immediate constant)

The execution times listed in Table 1 assume a 12MHz clock frequency. All of the arithmetic instructions execute in 1 $\mu$ s except the INC DPTR instruction, which takes 2 $\mu$ s, and the Multiply and Divide instructions, which take 4 $\mu$ s.

Note that any byte in the internal Data Memory space can be incremented without going through the Accumulator.

One of the INC instructions operates on the 16-bit Data Pointer. The Data Pointer is used to generate 16-bit addresses for external memory, so being able to increment it in one 16-bit operation is a useful feature.

The MUL AB instruction multiplies the Accumulator by the data in the B register and puts the 16-bit product into the concatenated B and Accumulator registers.

The DIV AB instruction divides the Accumulator by the data in the B register and leaves the 8-bit quotient in the Accumulator, and the 8-bit remainder in the B register.

Oddly enough, DIV AB finds less use in arithmetic "divide" routines than in radix conversions and programmable shift operations. An example of the use of DIV AB in a radix conversion will be given later. In shift operations, dividing a number by 2n shifts its n bits to the right. Using DIV AB to perform the division completes the shift in 4µs and leaves the B register holding the bits that were shifted out. The DA A instruction is for BCD arithmetic operations. In BCD arithmetic, ADD and ADDC instructions should always be followed by a DA A operation, to ensure that the result is also in BCD. Note that DA A will not convert a binary number to BCD. The DA A operation produces a meaningful result only as the second step in the addition of two BCD bytes.

MNEMONIC	OPERATION	ADDRESSING MODES		EXECUTION		
		DIR	IND	REG	IMM	TIME (μs)
ADD A, <byte></byte>	A = A + <byte></byte>	Х	Х	X	Х	1
ADDC A, <byte></byte>	$A = A + \langle byte \rangle + C$	Х	Х	X	Х	1
SUBB A, <byte></byte>	$A = A - \langle byte \rangle - C$	Х	X	X	Х	1
INC A	A = A + 1		Accumu	lator only		1
INC <byte></byte>	<byte> = <byte> + 1</byte></byte>	Х	Х	Х		1
INC DPTR	DPTR = DPTR + 1		Data Poi	inter only		2
DEC A	A = A - 1		Accumu	lator only	_	1
DEC <byte></byte>	<byte> = <byte> - 1</byte></byte>	Х	Х	X		1
MUL AB	$B:A = B \times A$		ACC an	d B only		4
DIV AB	$ \begin{array}{l} A = Int[A/B] \\ B = Mod[A/B] \end{array} $		ACC an	d B only		4
DA A	Decimal Adjust		Accumu	lator only		1

#### Table 1. 80C51 Arithmetic Instructions

## 80C51 family architecture

### Logical Instructions

Table 2 shows the list of 80C51 logical instructions. The instructions that perform Boolean operations (AND, OR, Exclusive OR, NOT) on bytes perform the operation on a bit-by-bit basis. That is, if the Accumulator contains 00110101B and byte contains 01010011B, then:

ANL A, <byte>

will leave the Accumulator holding 00010001B.

The addressing modes that can be used to access the <byte> operand are listed in Table 2.

The ANL A, <byte> instruction may take any of the forms:

- ANL A,7FH (direct addressing)
- ANL A,@R1 (indirect addressing)
- ANL A,R6 (register addressing)
- ANL A,#53H (immediate constant)

All of the logical instructions that are Accumulator-specific execute in 1 $\mu$ s (using a 12MHz clock). The others take 2 $\mu$ s.

Note that Boolean operations can be performed on any byte in the internal Data Memory space without going through the Accumulator. The XRL <byte>, #data instruction, for example, offers a quick and easy way to invert port bits, as in XRL P1, #OFFH.

If the operation is in response to an interrupt, not using the Accumulator saves the time and effort to push it onto the stack in the service routine.

The Rotate instructions (RL, A, RLC A, etc.) shift the Accumulator 1 bit to the left or right. For a left rotation, the MSB rolls into the LSB position. For a right rotation, the LSB rolls into the MSB position.

The SWAP A instruction interchanges the high and low nibbles within the Accumulator. This is a useful operation in BCD manipulations. For example, if the Accumulator contains a binary number which is known to be less than 100, it can be quickly converted to BCD by the following code:

MOVE	B,#10
DIV	AB
SWAP	А
ADD	A,B

Dividing the number by 10 leaves the tens digit in the low nibble of the Accumulator, and the ones digit in the B register. The SWAP and ADD instructions move the tens digit to the high nibble of the Accumulator, and the ones digit to the low nibble.

### **Data Transfers**

#### Internal RAM

Table 3 shows the menu of instructions that are available for moving data around within the internal memory spaces, and the addressing modes that can be used with each one. With a 12MHz clock, all of these instructions execute in either 1 or  $2\mu$ s.

The MOV <dest>, <src> instruction allows data to be transferred between any two internal RAM or SFR locations without going through the Accumulator. Remember, the Upper 128 bytes of data RAM can be accessed only by indirect addressing, and SFR space only by direct addressing.

Note that in 80C51 devices, the stack resides in on-chip RAM, and grows upwards. The PUSH instruction first increments the Stack Pointer (SP), then copies the byte into the stack. PUSH and POP use only direct addressing to identify the byte being saved or restored, but the stack itself is accessed by indirect addressing using the SP register. This means the stack can go into the Upper 128 bytes of RAM, if they are implemented, but not into SFR space.

The Upper 128 bytes of RAM are not implemented in the 80C51 nor in its ROMless or EPROM counterparts. With these devices, if the SP points to the Upper 128, PUSHed bytes are lost, and POPed bytes are indeterminate.

The Data Transfer instructions include a 16-bit MOV that can be used to initialize the Data Pointer (DPTR) for look-up tables in Program Memory, or for 16-bit external Data Memory accesses.

MNEMONIC	OPERATION		ADDRESS		s	EXECUTION
		DIR	IND	REG	ІММ	TIME (μs)
ANL A, <byte></byte>	A = A.AND. <byte></byte>	Х	Х	Х	Х	1
ANL <byte>,A</byte>	<byte> = <byte> .AND.A</byte></byte>	Х				1
ANL <byte>,#data</byte>	<byte> = <byte> .AND.#data</byte></byte>	Х				2
ORL A, <byte></byte>	A = A.OR. <byte></byte>	Х	Х	Х	Х	1
ORL <byte>,A</byte>	<byte> = <byte> .OR.A</byte></byte>	Х				1
ORL <byte>,#data</byte>	<byte> = <byte> .OR.#data</byte></byte>	Х				2
XRL A, <byte></byte>	A = A.XOR. <byte></byte>	Х	Х	Х	Х	1
XRL <byte>,A</byte>	<byte> = <byte> .XOR.A</byte></byte>	Х				1
XRL <byte>,#data</byte>	<byte> = <byte> .XOR.#data</byte></byte>	Х				2
CRL A	A = 00H		Accumu	lator only		1
CPL A	A = .NOT.A		Accumu	lator only		1
RL A	Rotate ACC Left 1 bit		Accumulator only			1
RLC A	Rotate Left through Carry		Accumu	lator only		1
RR A	Rotate ACC Right 1 bit		Accumulator only			1
RRC A	Rotate Right through Carry		Accumu	lator only		1
SWAP A	Swap Nibbles in A		Accumu	lator only		1

### Table 2. 80C51 Logical Instructions

MNEMONIC	OPERATION	ADDRESSING MODES		EXECUTION		
		DIR	IND	REG	IMM	TIME (μs)
MOV A, <src></src>	A = <src></src>	Х	Х	Х	Х	1
MOV <dest>,A</dest>	<dest> = A</dest>	Х	Х	Х		1
MOV <dest>,<src></src></dest>	<dest> = <src></src></dest>	Х	Х	Х	Х	2
MOV DPTR,#data16	DPTR = 16-bit immediate constant				Х	2
PUSH <src></src>	INC SP:MOV"@SP", <src></src>	Х				2
POP <dest></dest>	MOV <dest>,"@SP":DEC SP</dest>	Х				2
XCH A, <byte></byte>	ACC and <byte> exchange data</byte>	Х	Х	Х		1
XCHD A,@Ri	ACC and @Ri exchange low nibbles		Х			1

### Table 3. Data Transfer Instructions that Access Internal Data Memory Space

The XCH A, <byte> instruction causes the Accumulator and addressed byte to exchange data. The XCHD A, @Ri instruction is similar, but only the low nibbles are involved in the exchange.

To see how XCH and XCHD can be used to facilitate data manipulations, consider first the problem of shifting an 8-digit BCD number two digits to the right. Figure 11 shows how this can be done using direct MOVs, and for comparison how it can be done using XCH instructions. To aid in understanding how the code works, the contents of the registers that are holding the BCD number and the content of the Accumulator are shown alongside each instruction to indicate their status after the instruction has been executed.

After the routine has been executed, the Accumulator contains the two digits that were shifted out on the right. Doing the routine with direct MOVs uses 14 code bytes and  $9\mu$ s of execution time (assuming a 12MHz clock). The same operation with XCHs uses only 9 bytes and executes almost twice as fast.

To right-shift by an odd number of digits, a one-digit shift must be executed.

Figure 12 shows a sample of code that will right-shift a BCD number one digit, using the XCHD instruction. Again, the contents of the registers holding the number and of the Accumulator are shown alongside each instruction.

First, pointers R1 and R0 are set up to point to the two bytes containing the last four BCD digits. Then a loop is executed which

		2A	2B	2C	2D	2E	ACC
MOV	A,2EH	00	12	34	56	78	78
MOV	2EH,2DH	00	12	34	56	56	78
MOV	2CH.2BH	00	12	12	34 34	56	78
MOV	2BH,#0	00	00	12	34	56	78
		2A	2B	2C	2D	2E	ACC
CLR	Α	2A 00	2B 12	2C 34	2D 56	2E 78	ACC 00
CLR XCH	A A,2BH	2A 00 00	2B 12 00	2C 34 34	2D 56 56	2E 78 78 78	ACC 00 12
CLR XCH XCH XCH	A A,2BH A,2CH A 2DH	2A 00 00 00	2B 12 00 00	2C 34 34 12 12	2D 56 56 56 34	2E 78 78 78 78 78	ACC 00 12 34 56
CLR XCH XCH XCH XCH	A A,2BH A,2CH A,2DH A2EH	2A 00 00 00 00 00	2B 12 00 00 00 00	2C 34 34 12 12 12	2D 56 56 56 34 34	2E 78 78 78 78 56	ACC 00 12 34 56 78
CLR XCH XCH XCH XCH <b>B. Usin</b>	A A,2BH A,2CH A,2CH A,2DH A2EH <b>Ig XCHs: 9 byte</b>	2A 00 00 00 00 00 85, <b>5</b> μ <b>s</b>	2B 12 00 00 00 00	2C 34 34 12 12 12 12	2D 56 56 56 34 34	2E 78 78 78 78 78 56	ACC 00 12 34 56 78

Figure 11. Shifting a BCD Number Two Digits to the Right

leaves the last byte, location 2EH, holding the last two digits of the shifted number. The pointers are decremented, and the loop is repeated for location 2DH. The CJNE instruction (Compare and Jump if Not Equal) is a loop control that will be described later. The loop executed from LOOP to CJNE for R1 = 2EH, 2DH, 2CH, and 2BH. At that point the digit that was originally shifted out on the right has propagated to location 2AH. Since that location should be left with 0s, the lost digit is moved to the Accumulator.

#### External RAM

Table 4 shows a list of the Data Transfer instructions that access external Data Memory. Only indirect addressing can be used. The choice is whether to use a one-byte address, @Ri, where Ri can be either R0 or R1 of the selected register bank, or a two-byte address, @DPTR. The disadvantage to using 16-bit addresses if only a few k bytes of external RAM are involved is that 16-bit addresses use all 8 bits of Port 2 as address bus. On the other hand, 8-bit addresses allow one to address a few bytes of RAM, as shown in Figure 5, without having to sacrifice all of Port 2. All of these instructions execute in 2  $\mu$ s, with a 12MHz clock.

Note that in all external Data RAM accesses, the Accumulator is always either the destination or source of the data.

The read and write strobes to external RAM are activated only during the execution of a MOVX instruction. Normally these signals are inactive, and in fact if they're not going to be used at all, their pins are available as extra I/O lines.

			2A	2B	2C	2D	2E	ACC
	MOV MOV	- R1,#2EH R0,#2DH	00 00	12 12	34 34	56 56	78 78	XX XX
	loop for R	1 = 2EH:						
LOOP:	MOV XCHD SWAP MOV DEC DEC CJNE	A,@R1 A,@R0 A @R1,A R1 R0 R1,#2AH,LOO	00 00 00 00 00 00 P	12 12 12 12 12 12	34 34 34 34 34 34	56 58 58 58 58 58 58	78 78 67 67 67	78 76 67 67 67 67
	loop for R loop for R loop for R	1 = 2DH: 1 = 2CH: 1 = 2BH:	00 00 08	12 18 01	38 23 23	45 45 45	67 67 67	45 23 01
	CLR XCH	A A,2AH	08 00	01 01	23 23	45 45	67 67	00 08
								5000469

Figure 12. Shifting a BCD Number One Digit to the Right

ADDRESS WIDTH	MNEMONIC	OPERATION	EXECUTION TIME (μs)
8 bits	MOVX A,@Ri	Read external RAM @Ri	2
8 bits	MOVX @Ri,A	Write external RAM @ Ri	2
16 bits	MOVX A,@DPTR	Read external RAM @ DPTR	2
16 bits	MOVX @DPTR,A	Write external RAM @ DPTR	2

### Table 4. 80C51 Data Transfer Instructions that Access External Data Memory Space

### Table 5. 80C51 Lookup Table Read Instructions

MNEMONIC	OPERATION	EXECUTION TIME (µs)
MOVC A,@A+DPTR	Read program memory at (A + DPTR)	2
MOVC A,@A+PC	Read program memory at (A + PC)	2

#### Lookup Tables

Table 5 shows the two instructions that are available for reading lookup tables in Program Memory. Since these instructions access only Program Memory, the lookup tables can only be read, not updated.

If the table access is to external Program Memory, then the read strobe is  $\overline{\text{PSEN}}.$ 

The mnemonic is MOVC for "move constant." The first MOVC instruction in Table 5 can accommodate a table of up to 256 entries numbered 0 through 255. The number of the desired entry is loaded into the Accumulator, and the Data Pointer is set up to point to the beginning of the table. Then:

MOVC A,@A+DPTR

copies the desired table entry into the Accumulator.

The other MOVC instruction works the same way, except the Program Counter (PC) is used as the table base, and the table is accessed through a subroutine. First the number of the desired entry is loaded into the Accumulator, and the subroutine is called: MOV A.ENTRY NUMBER

MOV	A,ENTRY	<i>'</i>	Ν	UN
CALL	TABLE			

The subroutine "TABLE" would look like this: TABLE: MOVC A,@A+PC RET

The table itself immediately follows the RET (return) instruction in Program Memory. This type of table can have up to 255 entries, numbered 1 through 255. Number 0 cannot be used, because at the time the MOVC instruction is executed, the PC contains the address of the RET instruction. An entry numbered 0 would be the RET opcode itself.

### **Boolean Instructions**

80C51 devices contain a complete Boolean (single-bit) processor. The internal RAM contains 128 addressable bits, and the SFR space can support up to 128 addressable bits as well. All of the port lines are bit-addressable, and each one can be treated as a separate single-bit port. The instructions that access these bits are not just conditional branches, but a complete menu of move, set, clear, complement, OR, and AND instructions. These kinds of bit operations are not easily obtained in other architectures with any amount of byte-oriented software.

The instruction set for the Boolean processor is shown in Table 6. All bit accesses are by direct addressing.

Bit addresses 00H through 7FH are in the Lower 128, and bit addresses 80H through FFH are in SFR space.

Note how easily an internal flag can be moved to a port pin: MOV C,FLAG

MOV P1.0,C

In this example, FLAG is the name of any addressable bit in the Lower 128 or SFR space. An I/O line (the LSB of Port 1, in this case) is set or cleared depending on whether the flag bit is 1 or 0.

The Carry bit in the PSW is used as the single-bit Accumulator of the Boolean processor. Bit instructions that refer to the Carry bit as C assemble as Carry-specific instructions (CLR C, etc.). The Carry bit also has a direct address, since it resides in the PSW register, which is bit-addressable.

Note that the Boolean instruction set includes ANL and ORL operations, but not the XRL (Exclusive OR) operation. An XRL operation is simple to implement in software. Suppose, for example, it is required to form the Exclusive OR of two bits:

C = bit1 .XRL. bit2

The software to do that could be as follows:

	MOV	C,bit1
	JNB	bit2,OVER
	CPL	С
OVER:	(continu	ie)

First, bit1 is moved to the Carry. If bit2 = 0, then C now contains the correct result. That is, bit1 .XRL. bit2 = bit1 if bit2 = 0. On the other hand, if bit2 = 1, C now contains the complement of the correct result. It need only be inverted (CPL C) to complete the operation.

This code uses the JNB instruction, one of a series of bit-test instructions which execute a jump if the addressed bit is set (JC, JB, JBC) or if the addressed bit is not set (JNC, JNB). In the above case, bit2 is being tested, and if bit2 = 0, the CPL C instruction is jumped over.

JBC executes the jump if the addressed bit is set, and also clears the bit. Thus a flag can be tested and cleared in one operation. All the PSW bits are directly addressable, so the Parity bit, or the general purpose flags, for example, are also available to the bit-test instructions.

#### **Relative Offset**

The destination address for these jumps is specified to the assembler by a label or by an actual address in Program memory. However, the destination address assembles to a relative offset byte. This is a signed (two's complement) offset byte which is added to the PC in two's complement arithmetic if the jump is executed. The range of the jump is therefore -128 to +127 Program Memory bytes relative to the first byte following the instruction.

	MNEMONIC	OPERATION	EXECUTION TIME (µs)
ANL	C,bit	C = C.AND.bit	2
ANL	C,/bit	C = C.ANDNOT.bit	2
ORL	C,bit	C = C.OR.bit	2
ORL	C,/bit	C = C.ORNOT.bit	2
MOV	C,bit	C = bit	1
MOV	bit,C	bit = C	2
CLR	С	C = 0	1
CLR	bit	bit = 0	1
SETB	С	C = 1	1
SETB	bit	bit = 1	1
CPL	С	C = .NOT.C	1
CPL	bit	bit = .NOT.bit	1
JC	rel	Jump if C = 1	2
JNC	rel	Jump if $C = 0$	2
JB	bit,rel	Jump if bit = 1	2
JNB	bit,rel	Jump if bit = 0	2
JBC	bit,rel	Jump if bit = 1; CLR bit	2

### Table 6. 80C51 Boolean Instructions

### Table 7. Unconditional Jumps in 80C51 Devices

	MNEMONIC	OPERATION	EXECUTION TIME (µs)
JMP	addr	Jump to addr	2
JMP	@A+DPTR	Jump to A + DPTR	2
CALL	addr	Call subroutine at addr	2
RET		Return from subroutine	2
RETI		Return from interrupt	2
NOP		No operation	1

#### **Jump Instructions**

Table 7 shows the list of unconditional jumps with execution time for a 12MHz clock.

The table lists a single "JMP addr" instruction, but in fact there are three SJMP, LJMP, and AJMP, which differ in the format of the destination address. JMP is a generic mnemonic which can be used if the programmer does not care which way the jump is encoded.

The SJMP instruction encodes the destination address as a relative offset, as described above. The instruction is 2 bytes long, consisting of the opcode and the relative offset byte. The jump distance is limited to a range of -128 to +127 bytes relative to the instruction following the SJMP.

The LJMP instruction encodes the destination address as a 16-bit constant. The instruction is 3 bytes long, consisting of the opcode and two address bytes. The destination address can be anywhere in the 64k Program Memory space.

The AJMP instruction encodes the destination address as an 11-bit constant. The instruction is 2 bytes long, consisting of the opcode, which itself contains 3 of the 11 address bits, followed by another byte containing the low 8 bits of the destination address. When the instruction is executed, these 11 bits are simply substituted for the low 11 bits in the PC. The high 5 bits stay the same. Hence the destination has to be within the same 2k block as the instruction following the AJMP.

In all cases the programmer specifies the destination address to the assembler in the same way: as a label or as a 16-bit constant. The assembler will put the destination address into the correct format for the given instruction. If the format required by the instruction will not support the distance to the specified destination address, a "Destination out of range" message is written into the List file.

The JMP @A+DPTR instruction supports case jumps. The destination address is computed at execution time as the sum of the 16-bit DPTR register and the Accumulator. Typically, DPTR is set up with the address of a jump table. In a 5-way branch, for example, an integer 0 through 4 is loaded into the Accumulator. The code to be executed might be as follows:

MÕV	DPTR,#JUMP TABLE
MOV	A, INDEX_NUMBER
RL	А
JMP	@A+DPTR

The RL A instruction converts the index number (0 through 4) to an even number on the range 0 through 8, because each entry in the jump table is 2 bytes long:

JUMP TABLE:

AJMP	CASE 0
AJMP	CASE 1
AJMP	CASE 2
AJMP	CASE 3
AJMP	CASE 4

## 80C51 family architecture

Table 7 shows a single "CALL addr" instruction, but there are two of them, LCALL and ACALL, which differ in the format in which the subroutine address is given to the CPU. CALL is a generic mnemonic which can be used if the programmer does not care which way the address is encoded.

The LCALL instruction uses the 16-bit address format, and the subroutine can be anywhere in the 64k Program Memory space. The ACALL instruction uses the 11-bit format, and the subroutine must be in the same 2k block as the instruction following the ACALL.

In any case, the programmer specifies the subroutine address to the assembler in the same way: as a label or as a 16-bit constant. The assembler will put the address into the correct format for the given instructions.

Subroutines should end with a RET instruction, which returns execution to the instruction following the CALL.

RETI is used to return from an interrupt service routine. The only difference between RET and RETI is that RETI tells the interrupt control system that the interrupt in progress is done. If there is no interrupt in progress at the time RETI is executed, then the RETI is functionally identical to RET.

Table 8 shows the list of conditional jumps available to the 80C51 user. All of these jumps specify the destination address by the relative offset method, and so are limited to a jump distance of -128 to +127 bytes from the instruction following the conditional jump instruction. Important to note, however, the user specifies to the assembler the actual destination address the same way as the other jumps: as a label or a 16-bit constant.

There is no Zero bit in the PSW. The JZ and JNZ instructions test the Accumulator data for that condition.

The DJNZ instruction (Decrement and Jump if Not Zero) is for loop control. To execute a loop N times, load a counter byte with N and terminate the loop with a DJNZ to the beginning of the loop, as shown below for N = 10.

MOV COUNTER,#10 LOOP: (begin loop)

> (end loop) DJNZ COUNTER,LOOP (continue)

Table 8.	Conditional Jumps in 80C51 Devices

The CJNE instruction (Compare and Jump if Not Equal) can also be used for loop control as in Figure 12. Two bytes are specified in the operand field of the instruction. The jump is executed only if the two bytes are not equal. In the example of Figure 12, the two bytes were data in R1 and the constant 2AH. The initial data in R1 was 2EH. Every time the loop was executed, R1 was decremented, and the looping was to continue until the R1 data reached 2AH.

Another application of this instruction is in "greater than, less than" comparisons. The two bytes in the operand field are taken as unsigned integers. If the first is less than the second, then the Carry bit is set (1). If the first is greater than or equal to the second, then the Carry bit is cleared.

### **CPU Timing**

All 80C51 microcontrollers have an on-chip oscillator which can be used if desired as the clock source for the CPU. To use the on-chip oscillator, connect a crystal or ceramic resonator between the XTAL1 and XTAL2 pins of the microcontroller, and capacitors to ground as shown in Figure 13.

Examples of how to drive the clock with an external oscillator are shown in Figure 14. Note that in the NMOS devices (8051, etc.) the signal at the XTAL2 pin actually drives the internal clock generator. In the CMOS devices (80C51, etc.), the signal at the XTAL1 pin drives the internal clock generator. The internal clock generator defines the sequence of states that make up the 80C51 machine cycle.



Figure 13. Using the On-Chip Oscillator

MNEMONIC	OPERATION		ADDRESSING MODES			EXECUTION
		DIR	DIR IND REG IMM			TIME (μs)
JZ rel	Jump if $A = 0$		Accumu	2		
JNZ rel	Jump if A ≠ 0		Accumulator only			
DJNZ <byte>,rel</byte>	Decrement and jump if not zero	Х	X X			2
CJNE A, <byte>,rel</byte>	Jump if A ≠ <byte></byte>	Х	X X			2
CJNE <byte>,#data,rel</byte>	Jump if <byte> ≠ #data</byte>		X X			2



Figure 14. Using an External Clock

### **Machine Cycles**

A machine cycle consists of a sequence of 6 states, numbered S1 through S6. Each state time lasts for two oscillator periods. Thus a machine cycle takes 12 oscillator periods or  $1\mu$ s if the oscillator frequency is 12MHz.

Each state is divided into a Phase 1 half and a Phase 2 half. Figure 15 shows that fetch/execute sequences in states and phases for various kinds of instructions. Normally two program fetches are generated during each machine cycle, even if the instruction being executed doesn't require it. If the instruction being executed doesn't need more code bytes, the CPU simply ignores the extra fetch, and the Program Counter is not incremented.

Execution of a one-cycle instruction (Figures 15a and 15b) begins during State 1 of the machine cycle, when the opcode is latched into the Instruction Register. A second fetch occurs during S4 of the same machine cycle. Execution is complete at the end of State 6 of this machine cycle.

The MOVX instructions take two machine cycles to execute. No program fetch is generated during the second cycle of a MOVX instruction. This is the only time program fetches are skipped. The fetch/execute sequence for MOVX instructions is shown in Figure 15d.

The fetch/execute sequences are the same whether the Program Memory is internal or external to the chip. Execution times do not depend on whether the Program Memory is internal or external.

Figure 16 shows the signals and timing involved in program fetches when the Program Memory is external. If Program Memory is external, then the Program Memory read strobe PSEN is normally activated twice per machine cycle, as shown in Figure 16a. If an access to external Data Memory occurs, as shown in Figure 16b, two PSENs are skipped, because the address and data bus are being used for the Data Memory access.

Note that a Data Memory bus cycle takes twice as much time as a Program Memory bus cycle. Figure 16 shows the relative timing of the addresses being emitted at Ports 0 and 2, and of ALE and PSEN. ALE is used to latch the low address byte from P0 into the address latch.

When the CPU is executing from internal Program Memory, PSEN is not activated, and program addresses are not emitted. However, ALE continues to be activated twice per machine cycle and so it is available as a clock output signal. Note, however, that one ALE is skipped during the execution of the MOVX instruction.



Figure 15. State Sequence in 80C51 Family Devices



Figure 16. Bus Cycles in 80C51 Family Devices Executing from External Program Memory

# 80C51 family architecture

(MSB) (LSB)										
	ĒĀ	Х	Х	ES	ET1	EX1	ET0	EX0		
Syn	nbol	Positi	on F	Function						
ĒĀ		IE.7	I ii e i'	Disables all interrupts. If $\overline{EA} = 0$ , no interrupt will be acknowledged. If $\overline{EA} = 1$ , each interrupt source is individually enabled or disabled by setting or clearing its enable bit.						
		IE.6	F	Reserved	I.					
		IE.5	F	Reserved.						
ES		IE.4	E i i	Enables or disables the Serial Port interrupt. If $ES = 0$ , the Serial Port interrupt is disabled.						
ET1		IE.3	E i i	Enables or disables the Timer 1 Overflow interrupt. If $ET1 = 0$ , the Timer 1 interrupt is disabled.						
EX1		IE.2	E	Enables or disables External Interrupt 1. If EX1 = 0, External Interrupt 1 is disabled.						
ETC	)	IE.1	E i i	Enables or disables the Timer 0 Overflow interrupt. If $ETO = 0$ , the Timer 0 interrupt is disabled.						
EXC	)	IE.0	E	Enables or disables Exeternal Interrupt 0. If EX0 = 0, External Interrupt 0 is disabled.						
								SU00474		

(MSB) (LSB)								
	X	x	Х	PS	PT1	PX1	PT0	PX0
Sy	mbol	Positio	on Fu	Inction				
		IP.7	Re	eserved				
		IP.6	Re	eserved				
		IP.5	Re	eserved				
PS	3	IP.4	De lev pri	Defines the Serial Port interrupt priority level. PS = 1 programs it to the higher priority level.				
PT	1	IP.3	De lev pri	Defines the Timer 1 interrupt priority level. PT1 = 1 programs it to the higher priority level.				
PX	(1	IP.2	De lev pri	Defines the External Interrupt 1 priority level. PX1 = 1 programs it to the higher priority level.				
PT	0	IP.1	Er pri hiç	Enables or disables the Timer 0 Interrupt priority level. PT) = 1 programs it to the higher priority level.				
РХ	(0	IP.0	De lev pri	Defines the External Interrupt 0 priority level. PX0 = 1 programs it to the higher priority level.				
								SU00475
		Figure <sup>•</sup>	18. In <sup>.</sup>	terrupt	Priorit	v (IP) R	leaiste	r

Figure 17. Interrupt Enable (IE) Register





Figure 19. Interrupt Control System

## 80C51 family architecture

#### **Interrupt Structure**

The 80C51 and its ROMless and EPROM versions have 5 interrupt sources: 2 external interrupts, 2 timer interrupts, and the serial port interrupt.

What follows is an overview of the interrupt structure for the device. More detailed information for specific members of the 80C51 derivative family is provided in later chapters of this user's guide.

#### Interrupt Enables

Each interrupt source can be individually enabled or disabled by setting or clearing a bit in the SFR named IE (Interrupt Enable). This register also contains a global disable bit, which can be cleared to disable all interrupts at once. Figure 17 shows the IE register.

#### Interrupt Priorities

Each interrupt source can also be individually programmed to one of two priority levels by setting or clearing a bit in the SFR named IP (Interrupt Priority). Figure 18 shows the IP register. A low-priority interrupt can be interrupted by a high-priority interrupt, but not by another low-priority interrupt. A high-priority interrupt can't be interrupted by any other interrupt source.

If two interrupt requests of different priority levels are received simultaneously, the request of higher priority is serviced. If interrupt requests of the same priority level are received simultaneously, an internal polling sequence determines which request is serviced. Thus within each priority level there is a second priority structure determined by the polling sequence. Figure 19 shows how the IE and IP registers and the polling sequence work to determine which if any interrupt will be serviced.

In operation, all the interrupt flags are latched into the interrupt control system during State 5 of every machine cycle. The samples are polled during the following machine cycle. If the flag for an enabled interrupt is found to be set (1), the interrupt system generates an LCALL to the appropriate location in Program Memory, unless some other condition blocks the interrupt. Several conditions can block an interrupt, among them that an interrupt of equal or higher priority level is already in progress.

The hardware-generated LCALL causes the contents of the Program Counter to be pushed into the stack, and reloads the PC with the beginning address of the service routine. As previously noted (Figure 3), the service routine for each interrupt begins at a fixed location.

Only the Program Counter is automatically pushed onto the stack, not the PSW or any other register. Having only the PC automatically saved allows the programmer to decide how much time should be spent saving other registers. This enhances the interrupt response time, albeit at the expense of increasing the programmer's burden of responsibility. As a result, many interrupt functions that are typical in control applications toggling a port pin for example, or reloading a timer, or unloading a serial buffer can often be completed in less time than it takes other architectures to complete.

#### Simulating a Third Priority Level in Software

Some applications require more than two priority levels that are provided by on-chip hardware in 80C51 devices. In these cases, relatively simple software can be written to produce the same effect as a third priority level. First, interrupts that are to have higher priority than 1 are assigned to priority 1 in the Interrupt Priority (IP) register. The service routines for priority 1 interrupts that are supposed to be interruptable by priority 2 interrupts are written to include the following code:

PUSH IE MOV IE,#MASK CALL LABEL ...... (execute service routine) POP IE RET LABEL: RETI

As soon as any priority interrupt is acknowledged, the Interrupt Enable (IE) register is redefined so as to disable all but priority 2 interrupts. Then a CALL to LABEL executes the RETI instruction, which clears the priority 1 interrupt-in-progress flip-flop. At this point any priority 1 interrupt that is enabled can be serviced, but only priority 2 interrupts are enabled.

POPing IE restores the original enable byte. Then a normal RET (rather than another RETI) is used to terminate the service routine. The additional software adds  $10\mu s$  (at 12MHz) to priority 1 interrupts.